

# WAAT-Lookup Formulas 08-26-2016 Accounting Seminar

## Table of Contents

Lookup Functions: .....	2
VLOOKUP Function: .....	3
LOOKUP Function: .....	4
MATCH function .....	5
INDEX function .....	6
CHOOSE function .....	7
Difference between Exact Match and Approximate Match for Lookup functions: .....	8
Use wildcard characters .....	8
Data Mismatch .....	8
Example 1: Data Validation List. VLOOKUP, Exact Match. IF & ISBLANK functions. Goal: Create Invoice. ....	9
Example 2: How to use VLOOKUP Exact Match, LEFT, MID and SEARCH to do a "Partial Text Lookup. Learn about Data Mismatch. Goal: Lookup Product Price.....	9
Example 3: VLOOKUP with Approximate Match Lookup. Cell Reference for Column Index Number. Goal: Get Rating and Commission.....	10
Example 4: VLOOKUP with Approximate Match and MATCH functions with Exact Match. MATCH for Column Index Number. Goal: Retrieve Rating & Commission. ....	10
Example 5: Use MATCH to Compare Two Lists. Goal: Is an item in List 1 also in List 2? .....	11
Example 6: INDEX and MATCH functions to do a "Lookup Left". Goal: Given a flight range, lookup boomerang name in first column. ....	11
Example 7: INDEX and MATCH to find Vendor for Low Bid. INDEX array argument is a row. Goal: lookup vendor name for lowest bid.....	12
Example 8: How to use VLOOKUP & CHOOSE to do a lookup to multiple tables. Goal: Lookup Commission rate from Multiple Lookup tables. ....	12
Example 9: Multiple VLOOKUP functions. Goal: Calculate taxes from complex tax table. ....	13
Example 10: LOOKUP with lookup_value and lookup_vector and result_vector arguments. Approximate Match. Goal: "Lookup Adding" to total costs. ....	13
List of Keyboards: .....	14

## Lookup Functions:

### 1) VLOOKUP

- Does vertical lookup, trying to match a value in the first column of a lookup table and then retrieve an item from a subsequent column. Does Exact or Approximate Match. Since most lookup tables are vertical, VLOOKUP is the most commonly used lookup function.

### 2) LOOKUP

- Does vertical or horizontal lookup. Only does Approximate Match. Since this function can handle array operations and VLOOKUP cannot, it is often used for Lookup Array Formulas.

### 3) MATCH

- Returns the relative position of an item in a horizontal or vertical list (one-way array, either vertical or horizontal)

### 4) INDEX

- a lookup function that can do a two-way lookup or one-way lookup and even lookup a whole row or column. INDEX and MATCH together are the most versatile lookup functions.

### 5) CHOOSE

- Can choose items from a list based on an index number. This functions is useful for situations where you have multiple lookup tables.

## VLOOKUP Function:

- 1) V means Vertical
- 2) The goal of VLOOKUP is:
  - To go and get a value and deliver it to a cell or formula.
  - Try to match a value in the first column of a lookup table and then retrieve an item from a subsequent column and returns that item to the cell or formula.
- 3) VLOOKUP arguments: VLOOKUP ( **lookup\_value** , **table\_array** , **col\_index\_num** , **[range\_lookup]** )
  - **lookup\_value**:
    1. The value you want to look up and try to match against the items in the first column of the lookup table.
  - **table\_array**:
    1. This is the lookup table.
    2. The first column in the cell range must contain the lookup\_value
  - **col\_index\_num**:
    1. Column number in the lookup table that holds the value you want to return to the cell.
    2. If price is in the 4<sup>th</sup> column of the lookup table, use 4 to indicate that the 4<sup>th</sup> column holds the price.
  - **[range\_lookup]**:
    1. This argument tells VLOOKUP what type of lookup you are doing: Exact or Approximate.
    2. Exact Match:
      - i. Use FALSE or 0.
      - ii. VLOOKUP will do a linear search from the first item in the first column of the lookup table and keep searching until it finds an exact match.
      - iii. If there are duplicates, it only finds the first one.
      - iv. If it can't find a match, it returns an #N/A error.
    3. Approximate Match
      - i. Use TRUE or 1 or omitted.
      - ii. First column must be sorted ascending (biggest to smallest).
        1. Sorted either numerically or alphabetically
      - iii. Approximate Match is what you use when you are looking up:
        1. Tax Rates
        2. Commission Rates
        3. Decimal Grades
        4. Drainage Pipe Size
      - iv. Metaphor for understanding how Approximate Match works:
        1. It starts its search at the first item in first column, and looks at each one and when it bumps into first bigger value, it jumps back one row.
      - v. Technically, Approximate Match finds the largest value in the sorted first column that is less than or equal to lookup\_value by using a Binary search that works this way:
        1. Binary Search reduces search time because it repeatedly divides the table in half and checks the one in the middle.
        2. It doesn't have to check each one.
        3. Binary Search and is faster than Linear Search (Exact Match)
      - vi. If lookup\_value is less than 1st value in table VLOOKUP returns #N/A

## LOOKUP Function:

- 1) LOOKUP is THE original Lookup function from the first spreadsheet invented (VisiCalc). This function was invented before VLOOKUP.
- 2) Two possible uses for LOOKUP function. Screen Tip shows the arguments for the two possibilities:

```
LOOKUP(lookup_value, lookup_vector, [result_vector])  
LOOKUP(lookup_value, array)
```

- 3) What LOOKUP function does:
  - Goes and gets an item and brings it back to the cell or formula.
  - LOOKUP can do either Vertical or Horizontal lookup.
  - It can ONLY do Approximate Match.
- 4) **Approximate Match ONLY**
  - If you sort (smallest to biggest) the first column of the array or the lookup\_vector, you can force LOOKUP into doing Exact Match.
- 5) If you use “lookup\_value” and “array” arguments:
  - **array** is the lookup table, where the first column must be sorted (smallest to biggest)
  - When you use **array** as your lookup table, LOOKUP does either Vertical or Horizontal lookup
  - If the number of rows in the lookup table are greater than or equal to the number of columns, LOOKUP does vertical lookup
    1. When doing vertical lookup it always uses the last column as the column that contains the values it will potentially return to the cell or formula (this is the reason we do not need to put a column index number into LOOKUP).
  - If the number of columns in the lookup table are greater than the number of rows, LOOKUP does horizontal lookup
    1. When doing horizontal lookup it always uses the last row as the row that contains the values it will potentially return to the cell or formula (this is the reason we do not need to put a row index number into LOOKUP).
- 6) If you use “lookup\_value” and “lookup\_vector” and “result\_vector” arguments:
  - The **lookup\_vector** argument is like the MATCH function because it will deliver a relative position to the LOOKUP function that will be used to pick up an item from the **result\_vector** argument. Said a different way: LOOKUP will find the position of the “lookup\_value” in the “lookup\_vector” to find the relative position, and that retrieve an item from the “result\_vector” in that relative position.
  - These arguments can hold array operations without using Ctrl + Shift + Enter:
    1. The **lookup\_vector** argument in the LOOKUP function
    2. The **result\_vector** argument in the LOOKUP function
- 7) There are two good reasons we might want to use LOOKUP rather than VLOOKUP:
  - We are doing Approximate Match lookup and we want to enter fewer arguments than if we were to use VLOOKUP (LOOKUP doesn’t need column index number like VLOOKUP)
  - We are doing Approximate Match lookup and need to make an array calculation, which VLOOKUP can NOT do.

## MATCH function

### 1) What it does:

- Searches for an item in a list and returns the relative position of the item in the list.
- More specifically: Returns the relative position of an item in a horizontal or vertical list (one-way array, either vertical or horizontal)

### 2) MATCH(lookup\_value, lookup\_array, [match\_type])

#### • lookup\_value:

1. The value you want to look up.
2. The value you want to look up must be in the range of cells you specify in lookup\_array.

#### • lookup\_array:

1. The range of cells being searched.
2. Must be a one-way array, either:
  - i. Vertical (one column with one or more rows)
  - ii. Horizontal (one row with one or more columns)

#### • [match\_type]:

1. Optional. The numbers: -1, 0, or 1.
2. The match\_type argument specifies how Excel matches lookup\_value with values in lookup\_array.
3. The default value for this argument is 1

#### 4. 1 or omitted

- i. MATCH finds the largest value that is less than or equal to lookup\_value. The values in the lookup\_array argument must be placed in ascending order.
- ii. Works just like VLOOKUP Approximate Match

#### 5. 0

- i. MATCH finds the first value that is exactly equal to lookup\_value. The values in the lookup\_array argument can be in any order.
- ii. Works just like VLOOKUP Exact Match

#### 6. -1

- i. MATCH finds the smallest value that is greater than or equal to lookup\_value. The values in the lookup\_array argument must be placed in descending order.

#### 7. Another alternative way to describe this argument:

- i. **[match\_type]** tells the MATCH what sort of lookup to do:
- ii. 1 or empty = approximate match; table sorted ascending; first bigger value bumped into then jump back one position, if value is smaller than first item returns #N/A, if bigger than last it returns last value
- iii. 2 = extract match, if duplicates, it finds first one only, can't find it it shows #N/A
- iv. -1 = approximate match; table sorted descending; first smaller value bumped into then jump back one position, if value is bigger than first item returns #N/A, if smaller than last it returns last value

## INDEX function

- 1) Two possible uses for INDEX function:

```
INDEX(array, row_num, [column_num])  
INDEX(reference, row_num, [column_num], [area_num])
```

- 2) We will ONLY use the first one with the arguments: **array**, **row-num** and **[column\_num]**.

- 3) What it does:

- INDEX and MATCH together are the most versatile lookup functions because they can do anything that VLOOKUP can do, plus much more such as:
  1. It can do two-way lookup: where we are retrieving an item from a table at the intersection of a row header and column header.
  2. It can do one-way lookup on a column of values (column filled with rows)
  3. It can do one-way lookup on a row of values (row filled with columns)
  4. It can look up a whole column of values
  5. It can look up a whole row of values

- 4) Arguments:

- **array:**

1. A two dimensional table (both a row and a column).  
or
2. A one dimensional table (row or column). The range/array containing the values you want to look up.

- **row\_num**

1. Tells INDEX from which row to retrieve the item.
2. If you put a 0 (zero) or omitted, all the rows are returned; this is how you “lookup a whole column”.

- **column\_num**

1. Tells INDEX from which column to retrieve the item.
2. If you put a 0 (zero) or omitted, all the columns are returned; this is how you “lookup a whole row”.

- 5) When INDEX looks up a row or column, it returns a range of values, not an array. This is why we can add with SUM.

## CHOOSE function

- 1) CHOOSE is a lookup function that can return “things” to a cell or a formula
- 2) "Things" you are allowed to lookup and return to a cell or formula:
  - Text
  - Numbers
  - Formulas
  - Functions
  - Cell References
  - Ranges
  - Defined Names
  - Array Constants
- 3) Function: **=CHOOSE(index\_num, value1,value2, ...)**
- 4) Arguments:
  - index\_num
    1. Must be an index number like: 1,2,3,4,5,...
  - value1, value2,...
    1. These are the “things” you want to return to cell or formula
    2. You actually store the “things” that want to lookup in the value1, value2 arguments.
      - i. Unlike other lookup functions, CHOOSE requires that you enter the "things" into the function individually

## Difference between Exact Match and Approximate Match for Lookup functions:

- 1) Exact Match:
  - Does a linear search from the first item and keep searching until it finds an exact match.
  - If there are duplicates, it only finds the first one.
  - If it can't find a match, it returns an #N/A error.
- 2) Approximate Match
  - Range must be sorted ascending (biggest to smallest).
    1. Sorted either numerically or alphabetically
  - Approximate Match is what you use when you are looking up:
    1. Tax Rates
    2. Commission Rates
    3. Decimal Grades
    4. Drainage Pipe Size
  - Approximate match finds the largest value that is less than or equal to lookup\_value.
  - Metaphor for understanding how it works:
    1. It starts its search at the first item, and looks at each item, one at a time and when it bumps into first bigger value, it jumps back one position.
  - Binary search is how it really works.
    1. Binary Search and is faster than Linear Search (Exact Match)
    2. Binary Search reduces search time because it repeatedly divides the table or range in half and checks the one in the middle to help reduce calculation time.
    3. It doesn't have to check each one.
  - If lookup\_value is less than 1st value in table or range it will return an #N/A

## Use wildcard characters

- 3) If you are doing an Exact Match and the lookup\_value is text, you can use the wildcard characters in lookup\_value.
  - Wildcard Characters:
    1. Question mark (?)
      - i. A question mark matches any single character.
    2. Asterisk (\*)
      - i. An asterisk matches any sequence of characters (zero or more characters).
  - If you want to find an actual question mark or asterisk, type a tilde (~) in front of the character.

## Data Mismatch

- 1) In order for data to be matched as equivalent, the two data types must be the same.
  - To convert Text Numbers back to number, you can use any math operation, such as:
    1. Add Zero: +0
    2. Double Negative: --
  - To Convert Numbers to Text Numbers you can Join a Zero Length Text String, such as: &""



## Example 1: Data Validation List. VLOOKUP, Exact Match. IF & ISBLANK functions. Goal: Create Invoice.

	A	B	C	D
1	<b>Example 1: Data Validation List. VLOOKUP, Exact Match. IF &amp; ISBLANK functions. Goal: Create Invoice.</b>			
2				
3	Boomerang	Part Number	Flight Range (meters)	Price
4	Bellen	1000-165-B100	25	\$26.95
5	Carlota	1001-540-C101	20	\$28.95
6	Majestic Beaut	1002-394-M102	35	\$31.95
7	Quad	1003-307-Q103	20	\$35.95
8	Sunshine	1004-848-S104	30	\$18.95
9	Sunset	1005-155-S105	40	\$20.95
10	Tri-Fly	1006-552-T106	1	\$4.95
11	Outdoor Tri-Fly	1007-634-O107	5	\$8.95
12				
13				
14	<b>Invoice Example:</b>			
15	Product	Units	Price	Total
16	Bellen	5	=IF(ISBLANK(B16),"",VLOOKUP(A16,\$A\$4:\$D\$11,4,0))	
17	Sunshine	2	18.95	37.9
18	Quad	10	35.95	359.5
19				
20				
21				
22			<b>Subtotal</b>	532.15

## Example 2: How to use VLOOKUP Exact Match, LEFT, MID and SEARCH to do a "Partial Text Lookup. Learn about Data Mismatch. Goal: Lookup Product Price.

	A	B	C	D	E	F
1	<b>Example 2: How to use VLOOKUP Exact Match, LEFT, MID and SEARCH to do a "Partial Text Lookup. Goal: Lookup Product Price.</b>					
2						
3	ID	Price	Price "Data Mismatch"		ID	Price
4	Bellen-234-B25R	=VLOOKUP(LEFT(A4,SEARCH("-",A4)-1),SE\$4:\$F\$6,2,0)				\$26.00
5	Carlota-345-C20R	\$23.00	\$23.00		Carlota	\$23.00
6	Quad-765-Q20L	\$36.00	\$36.00		Quad	\$36.00
7						
8					ID	Price
9					234	\$26.00
10					345	\$23.00
11					765	\$36.00

And

	A	B	C	D	E	F
1	<b>Example 2: How to use VLOOKUP Exact Match, LEFT, MID and SEARCH to do a "Partial Text Lookup. Goal: Lookup Product Price.</b>					
2						
3	ID	Price	Price "Data Mismatch"		ID	Price
4	Bellen-234-B25R	\$26.00	=VLOOKUP(MID(A4,SEARCH("-",A4)+1,3)+0,SE\$9:\$F\$11,2,0)			
5	Carlota-345-C20R	\$23.00	\$23.00		Carlota	\$23.00
6	Quad-765-Q20L	\$36.00	\$36.00		Quad	\$36.00
7						
8					ID	Price
9					234	\$26.00
10					345	\$23.00
11					765	\$36.00

Example 3: VLOOKUP with Approximate Match Lookup. Cell Reference for Column Index Number. Goal: Get Rating and Commission.

	A	B	C	D
1	<b>Example 3: VLOOKUP with Approximate Match Lookup. Cell Reference for Column Index Number. Goal: Get Rating</b>			
2				
3	Sales Amount	Rating	Commission Paid	
4	\$0.00	Sub Par	\$0.00	
5	\$1,000.00	Par	\$20.00	
6	\$2,500.00	Above Par	\$100.00	
7	\$7,000.00	Very Good	\$250.00	
8	\$10,000.00	Excellent	\$700.00	
9				
10			2	3
11	Employee	Sales	Rating	Commission Paid
12	Sioux	\$7,598.00	=VLOOKUP(\$B12,\$A\$4:\$C\$8,C\$10)	
13	Kim	\$68.00	Sub Par	\$0.00
14	Gigi	\$15,980.00	Excellent	\$700.00
15	Franny	\$2,499.99	Par	\$20.00
16				

Example 4: VLOOKUP with Approximate Match and MATCH functions with Exact Match. MATCH for Column Index Number. Goal: Retrieve Rating & Commission.

	A	B	C	D
1	<b>Example 4: VLOOKUP with Approximate Match and MATCH functions with Exact Match. MATCH for Column Index Number.</b>			
2	MATCH function is a lookup function.			
3	MATCH function returns the relative position of an item in a list.			
4	This method is dynamic: 1) columns can be in any order and 2) If you change input formula updates.			
5				
6	Sales Amount	Rating	Commission Paid	
7	\$0.00	Sub Par	\$0.00	
8	\$1,000.00	Par	\$20.00	
9	\$2,500.00	Above Par	\$100.00	
10	\$7,000.00	Very Good	\$250.00	
11	\$10,000.00	Excellent	\$700.00	
12				
13			2	3
14	Employee	Sales	Rating	Commission Paid
15	Sioux	\$7,598.00	=VLOOKUP(\$B15,\$A\$7:\$C\$11,MATCH(C\$14,\$A\$6:\$C\$6,0))	
16	Kim	\$68.00	Sub Par	0
17	Gigi	\$15,980.00	Excellent	700
18	Franny	\$2,499.99	Par	20

### Example 5: Use MATCH to Compare Two Lists. Goal: Is an item in List 1 also in List 2?

	A	B	C	D	E
1	<b>Example 5:</b> Use MATCH to Compare Two Lists. <b>Goal:</b> Is an item in List 1 also in List 2?				
2	<b>Notes:</b>				
3	MATCH: reports relative position of item in a list				
4	F4 key adds \$ signs to lock range (make absolute).				
5	ISNUMBER: reports TRUE when it sees a number				
6					
7	<b>Has prospective customer made it into our master list due to our sales phone calls?</b>				
8			<b>List 2 = Customers we have made sales calls to.</b>		
9	<b>Master Customer List</b>		<b>Prospective Customers</b>		
10	<b>List 1</b>		<b>List 2</b>	<b>Is item in List 2 in List 1?</b>	
11	Fran's Produce		Julie's Produce	=ISNUMBER(MATCH(C11,\$A\$11:\$A\$29,0))	
12	Produce Fast And Fresh		Health Choice Fruit	ISNUMBER(value)	
13	Veggies And Fruit Delight		Fresh Delights	TRUE	

### Example 6: INDEX and MATCH functions to do a "Lookup Left". Goal: Given a flight range, lookup boomerang name in first column.

	A	B	C	D
1	<b>Example 6:</b> INDEX and MATCH functions to do a "Lookup Left". Goal: Given a flight range, lookup boomerang name			
2	INDEX function: Can do many looks that VLOOKUP cannot			
3	array argument = values that you want to retrieve. array can be a table, a column, or a row.			
4	row_num argument = tells INDEX what row to look at to retrieve a value. If the array in a one-way array, row			
5	column_num argument tells INDEX what column to look at to retrieve a value. If you put 0 or omitted, it will			
6				
7	Use INDEX as a substitute for VLOOKUP when the value you want to return is to the left of the lookup column.			
8	<b>Product</b>	<b>Price</b>	<b>Description</b>	
9	V Range	\$12.95	Flying Range is 10	
10	Quad	\$39.95	Flying Range is 20	
11	Weighted Sunset	\$40.00	Flying Range is 50	
12	Weighted MB	\$45.00	Flying Range is 60	
13	S Rang	\$65.00	Flying Range is 70	
14	Long D	\$69.00	Flying Range is 80	
15	Long D Squared	\$100.00	Flying Range is 85	
16	Dan's Special	\$110.00	Flying Range is 110	
17	Manu Magic	\$165.00	Flying Range is 160	
18				
19	1-Way Lookup Vertical: Lookup Column to Left			
20		<b>Description</b>	<b>Product</b>	
21		Flying Range is 20	=INDEX(A9:A17,MATCH(B21,C9:C17,0))	

Example 7: INDEX and MATCH to find Vendor for Low Bid. INDEX array argument is a row. Goal: lookup vendor name for lowest bid.

	A	B	C	D	E	F	G	H	I
1	<b>Example 7: INDEX and MATCH to find Vendor for Low Bid. INDEX array argument is a row. Goal: lookup vendor name for lowest bid.</b>								
2									
3	<b>Venders and bids:</b>								
4	Crank'ys	Bay Air	Compressor R Us	Mech-Aid	Han's Shop	Low Bid	Vendor for Low Bid		
5	\$38.99	\$48.60	\$43.53	\$40.08	\$47.92	\$38.99	=INDEX(\$A\$4:\$E\$4,MATCH(F5,A5:E5,0))		
6	\$57.68	\$31.80	\$52.78	\$31.42	\$55.19	\$31.42	Mech-Aid		
7	\$53.32	\$32.64	\$37.69	\$48.29	\$41.59	\$32.64	Bay Air		
8	\$35.20	\$40.55	\$32.65	\$36.81	\$41.14	\$32.65	Compressor R Us		
9	\$56.72	\$47.16	\$36.42	\$49.56	\$39.25	\$36.42	Compressor R Us		

Example 8: How to use VLOOKUP & CHOOSE to do a lookup to multiple tables. Goal: Lookup Commission rate from Multiple Lookup tables.

	A	B	C	D	E	F	G	H	I
1	<b>Example 8: How to use VLOOKUP &amp; CHOOSE to do a lookup to multiple tables. Goal: Lookup Commission rate from Multiple Lookup tables.</b>								
2									
3	CHOOSE functions allows you to lookup things like different lookup tables.								
4	=CHOOSE(index_num, value1,value2, ...)								
5	index_num is a number like 1, 2, 3, 4								
6	value1 = lookup table 1								
7	value2 = lookup table 2								
8	value3 = lookup table 3								
9	If index_num is 2, then the lookup table 2 will be returned by CHOOSE to the VLOOKUP function								
10									
11	<b>VLOOKUP &amp; CHOOSE with 3 lookup tables</b>								
12						ABC			
13	Product	Units Sold	Table	Commission Rate		Units Sold	Commission Rate		
14	ABC	431	1	4.00%		0	1.00%		
15	EDR	65	2	1.00%		100	2.00%		
16	EDS	563	3	4.00%		200	4.00%		
17	ABC	493	1	4.00%		500	6.00%		
18	EDS	188	3	2.00%					
19						EDR			
20						Units Sold	Commission Rate		
21	ABC	1		Commission Rate		0	1.00%		
22	EDR	2		=VLOOKUP(B14,CHOOSE(VLOOKUP(A14,\$A\$21:\$B\$23,2,0),\$F\$13:\$G\$16,\$F\$20:\$G\$23,\$F\$27:\$G\$30),2)		200	2.00%		
23	EDS	3		1.00%		400	6.00%		
24				4.00%					
25				4.00%		EDS			
26				2.00%		Units Sold	Commission Rate		

### Example 9: Multiple VLOOKUP functions. Goal: Calculate taxes from complex tax table.

	A	B	C	D	E	F
1	<b>Example 9: Multiple VLOOKUP functions. Goal: Calculate taxes from complex tax table.</b>					
2	<b>Tax Example:</b>					
3		<b>Married</b>				
4	<b>Lookup Value</b>	<b>Over --</b>	<b>But not over --</b>	<b>Rate</b>	<b>Tax from Previous brackets</b>	<b>Rule</b>
5	\$0	\$0	\$1,313	0%	\$0.00	Zero Tax
6	\$1,313.01	\$1,313	\$2,038	10%	\$0.00	10% of excess over \$1,313
7	\$2,038.01	\$2,038	\$6,304	15%	\$72.50	\$72.50 + 15% of excess over \$2,038
8	\$6,304.01	\$6,304	\$9,844	25%	\$712.40	\$712.40 + 25% of excess over \$6,304
9	\$9,844.01	\$9,844	\$18,050	28%	\$1,597.40	\$1,597.40 + 28% of excess over \$9,844
10	\$18,050.01	\$18,050	\$31,725	33%	\$3,895.08	\$3,895.08 + 33% of excess over \$18,050
11	\$31,725.01	\$31,725		35%	\$8,407.83	\$8,407.83 + 35% of excess over \$31,725
12						
13	<b>Taxable Earnings</b>	\$15,896.00				
14	<b>Tax from Previous brackets</b>	\$1,597.40				
15	<b>Tax Rate</b>	28.00%				
16	<b>Amount already taxed</b>	\$9,844.00				
17	<b>Amount to tax in this bracket</b>	\$6,052.00				
18	<b>Total Tax</b>	\$3,291.96				
19						
20	<b>Total Tax</b>	=VLOOKUP(B13,A5:E11,5)+ROUND((B13-VLOOKUP(B13,A5:E11,2))*VLOOKUP(B13,A5:E11,4),2)				
21		VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])				

### Example 10: LOOKUP with lookup\_value and lookup\_vector and result\_vector arguments. Approximate Match. Goal: "Lookup Adding" to total costs.

	A	B	C	D	E	F	G	H	I	J	K
1	<b>Example 10: LOOKUP with lookup_value and lookup_vector and result_vector arguments. Approximate Match. Goal: "Lookup Adding" to total costs.</b>										
2											
3	<b>Reason we use LOOKUP: LOOKUP lookup_value argument can do a function argument array operation, but VLOOKUP lookup_value can NOT.</b>										
4											
5	A company that sells rolls of fence needs to calculate total revenue from a list of customer transactions and a pricing table.										
6	Given the transactional data table with the number of rolls of fence sold for each transaction and the pricing table,										
7	calculate total revenue and count the number of transactions that fell into each pricing category.										
8											
9	<b>Data set with number of rolls of fence sold for each transaction:</b>						<b>Pricing Table:</b>				
10											
11	<b>Date</b>	<b>Number of Rolls</b>	<b>Price</b>	<b>Revenue</b>		<b>Total Revenue Helper Column</b>	<b>Total Revenue Single Cell Formula</b>		<b>Units From</b>	<b>Units To</b>	<b>Price per Roll (100 feet)</b>
12	9/16/2015	540	\$168.00	\$90,720.00		\$11,852,436.00	=SUMPRODUCT(LOOKUP(B12:B183,I12:K15),B12:B183)				\$198.00
13	9/15/2015	360	\$168.00	\$60,480.00					144	288	\$187.00
14	9/16/2015	12	\$198.00	\$2,376.00					289	577	\$168.00
15	9/15/2015	468	\$168.00	\$78,624.00					578	and up	\$152.00
16	9/17/2015	48	\$198.00	\$9,504.00							

## List of Keyboards:

- 1) **Esc Key:**
  - i. Closes Backstage View (like Print Preview).
  - ii. Closes most dialog boxes.
  - iii. If you are in Edit mode in a Cell, Esc will revert back to what you had in the cell before you put the Cell in Edit mode.
- 2) **F2 Key** = Puts formula in Edit Mode and shows the rainbow colored Range Finder.
- 3) **SUM Function: Alt + =**
- 4) **Ctrl + Shift + Arrow** = Highlight column (Current Region).
- 5) **Ctrl + Backspace** = Jumps back to Active Cell
- 6) **Ctrl + Z** = Undo.
- 7) **Ctrl + Y** = Undo the Undo.
- 8) **Ctrl + C** = Copy.
- 9) **Ctrl + X** = Cut.
- 10) **Ctrl + V** = Paste.
- 11) **Ctrl + PageDown** = expose next sheet to right.
- 12) **Ctrl + PageUp** = expose next sheet to left.
- 13) **Ctrl + 1** = Format Cells dialog box, or in a chart it opens Format Chart Element Task Pane.
- 14) **Ctrl + Arrow**: jumps to the bottom of the "Current Region", which means it jumps to the last cell that has data, right before the first empty cell.
- 15) **Ctrl + Home** = Go to Cell A1.
- 16) **Ctrl + End** = Go to last cell used.
- 17) Alt keyboards are keys that you hit in succession. Alt keyboards are keyboards you can teach yourself by hitting the Alt key and looking at the screen tips.
  - i. Create PivotTable dialog box: **Alt, N, V**
  - ii. Page Setup dialog box: **Alt, P, S, P**
  - iii. Keyboard to open Sort dialog box: **Alt, D, S**
- 18) **ENTER** = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell DOWN.
- 19) **CTRL + ENTER** = When you are in Edit Mode in a Cell, it will put thing in cell and keep cell selected.
- 20) **TAB** = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell RIGHT.
- 21) **SHIFT + ENTER** = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell UP.
- 22) **SHIFT + TAB** = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell LEFT.
- 23) **Ctrl + T** = Create Excel Table (with dynamic ranges) from a Proper Data Set.
  - i. Keyboard to name Excel Table: **Alt, J, T, A**
  - ii. **Tab** = Enter Raw Data into an Excel Table.
- 24) **Ctrl + Shift + ~ ( ` )** = General Number Formatting Keyboard.
- 25) **Ctrl + ;** = Keyboard for hardcoding today's date.
- 26) **Ctrl + Shift + ;** = Keyboard for hardcoding current time.
- 27) **Arrow Key** = If you are making a formula, Arrow key will "hunt" for Cell Reference.
- 28) **Ctrl + B** = Bold the Font
- 29) **Ctrl + \* (on Number Pad)** or **Ctrl + Shift + 8** = Highlight Current Table.
- 30) **Alt + Enter** = Add Manual Line Break (Word Wrap)
- 31) **Ctrl + P** = Print dialog Backstage View and Print Preview
- 32) **F4 Key** = If you are in Edit mode while making a formula AND your cursor is touching a particular Cell Reference, F4 key will toggle through the different Cell References:
  - i. **A1** = Relative
  - ii. **\$A\$1** = Absolute or "Locked"

- iii. **A\$1** = Mixed with Row Locked (Relative as you copy across the columns AND Locked as you copy down the rows)
  - iv. **\$A1** = Mixed with Column Locked (Relative as you copy down the rows AND Locked as you across the columns)
- 33) **Ctrl + Shift + 4** = Apply Currency Number Formatting
- 34) **Tab key** = When you are selecting a Function from the Function Drop-down list, you can select the function that is highlighted in blue by using the Tab key.
- 35) **F9 Key** = To evaluate just a single part of formula while you are in edit mode, highlight part of formula and hit the F9 key.
- i. If you are creating an Array Constant in your formula: Hit F9.
  - ii. If you are evaluating the formula element just to see what that part of the formula looks like,  
REMEMBER: to Undo with Ctrl + Z.
- 36) **Alt, E, A, A** = Clear All (Content and Formatting)
- 37) Evaluate Formula One Step at a Time Keyboard: **Alt, M, V**
- 38) Keyboard to open Sort dialog box: **Alt, D, S**
- 39) **Ctrl + Shift + L** = Filter (or **Alt, D, F, F**) = Toggle key for Filter Drop-down Arrows
- 40) **Ctrl + N** = Open New File
- 41) **F12** = Save As (Change File Name, Location, File Type)
- 42) Import Excel Table into Power Query Editor: **Alt, A, P, T**
- 43) **Ctrl + 1 (When Chart element in selected)**: Open Task Pane for Chart Element
- 44) **F4 Key** = If you are in Edit mode while making a formula AND your cursor is touching a particular Cell Reference, F4 key will toggle through the different Cell References:
- i. **A1** = Relative
  - ii. **\$A\$1** = Absolute or "Locked"
  - iii. **A\$1** = Mixed with Row Locked (Relative as you copy across the columns AND Locked as you copy down the rows)
  - iv. **\$A1** = Mixed with Column Locked (Relative as you copy down the rows AND Locked as you across the columns)
- 45) Keyboard to open Scenario Manager = **Alt, T, E**
- 46) **Ctrl + Tab** = Toggle between Excel Workbook File Windows
- 47) **Ctrl + Shift + F3** = Create Names From Selection
- 48) **Ctrl + F3** = open Name Manager
- 49) **F3** = Paste Name or List of Names
- 50) **Alt + F4** = Close Active Window
- 51) **Window Key + Up Arrow** = Maximize Active Window
- 52) **Ctrl + Shift + Enter** = Keystroke to enter Array Formulas that: 1) have a function argument that requires it, or 2) whether or not you are entering the Resultant Array into multiple cells simultaneously.
- 53) **Ctrl + /** = Highlight current Array